

PROCEDURE FOR SETTING SYN2 FREQUENCY THROUGH MCM
venkat

1.0 Introduction:

SYN2 is the acronym for the "First LO SYNthesiser #2", used in GMRT for generating phase-coherent LO in the range of 600 to 1600 MHz. This unit has been recently prototyped and will be commissioned at C1 and C2 on an experimental basis next week for field evaluation. This note summarises the procedure to be followed for controlling SYN2 through ONLINE using MCM3.

NOTE THAT THE EXISTING FACILITIES FOR CONTROLLING SYN1 OR FRONT-END THROUGH MCM2 WILL NOT BE DISTURBED.

2.0 The Program:

The GWBASIC program to generate a dos runfile SYN2.DAT is given in Appendix 1. The program setx.c in magha at Khodad has been updated to include the new facility.

It may be noted that a simple equation for coarse control of YIG frequency F in the form

Synthesised frequency (MHz) = YIGDAC bit * 25 MHz/bit + 200
has been used in the program. This is a rough approximation, which will be refined later when more units are built.

Because of the above approximation, while the system will lock at a majority of frequencies in the range of 600 to 1600 MHz in 5 MHz step, the same CANNOT BE GUARANTEED for all frequencies. The procedure as explained in 4.0 below has to be used to ensure that SYN2 is in lock and the system is usable.

3.0 Monitoring:

SYN2 will have the facility for monitoring CRUCIAL parameters which will be mapped to the following channels of MCM3 in the commissioned antennas:

Digital lock indication	:	Channel 24
Coarse lock voltage (CLV)	:	Channel 25
Analog lock indication	:	Channel 24
Fine lock voltage (FLV)	:	Channel 27

The Digital Lock indication is a FAIL-PROOF indication for lock. Channel 24 should read < 180 counts (Typically 130) under lock and > 180 counts (Typically 240) when out-of-lock.

Interpretation of channels 25, 26 and 27 will be done by the Engineers.

Users are requested to ensure that contents of addresses 24 to 27 is logged whenever SYN2 is used.

4.0 Procedure to ensure SYN2 is in lock:

(a) Edit SETUP.TXT file and run setx.exe in magha. Execute the runfile LO.SET through ONLINE and monitor address 25.

(b) If < 180 counts, SYN2 is in lock AND THE SYSTEM CAN BE USED FOR OBSERVATION.

(c) If not, note down the value of V3\$ and V4\$ in the command issued (say, 2Fxx AFxx) and reissue a 32 bit command to MCM3 with xx DECREMENTED or INCREMENTED by 1. The system will now be in lock.

(d) If still not in lock, CONTACT ME!!!

Appendix 1.

```
10 'PROGRAM for SYN2 bit pattern generation venkat 20/09/95
20 CLS:LOCATE 5,1:INPUT "FREQUENCY OF SYN2 TO SYNTHESISE :: ",F
30 IF ((F < 600) OR (F > 1600)) GOTO 20
40 TMP=F:WHILE TMP > 0:TMP=TMP-5:WEND
50 IF (TMP <> 0) GOTO 20
60 'IN STATEMENT 80, ST STANDS FOR STEP SIZE AND YIGCS, FOR YIG
70 'COARSE SENSITIVITY.
80 YIGCS=25:ST=5
90 F1=F/ST
100 MDEC=F1-10*INT(F1/10) : MNUM=15-MDEC
110 NDEC=(INT(F1/10)) : NNUM=255-NDEC+1
120 BX=NNUM*16
130 ACUM=BX+MNUM
140 L$="4":T$="C"
150 ACUM=4095-ACUM
160 IF (F < 850) THEN L$="40":T$="C0"
170 V1$=L$+HEX$(ACUM)
180 V2$=T$+HEX$(ACUM)
220 BITSET=INT((F-200)/YIGCS)
230 ACUM=BITSET+3840
240 V3$="2"+HEX$(ACUM)
250 V4$="A"+HEX$(ACUM)
260 V5$="0258":V6$="8258"
270 V7$="1008":V8$="9008"
280 OPEN "SYN2.DAT" FOR OUTPUT AS #1
290 PRINT "*FREQ:: ";F;" MHz; STP:: ";ST;" MHz; VCO :: YIG"
300 'PRINT ":3"
310 PRINT USING "\ \";V1$;:PRINT " ";:PRINT USING "\ \";V2$;
320 PRINT " ";:PRINT USING "\ \";V3$;
330 PRINT " ";:PRINT USING "\ \";V4$
340 PRINT USING "\ \";V5$;:PRINT " ";:PRINT USING "\ \";V6$;
350 PRINT " ";:PRINT USING "\ \";V7$;
360 PRINT " ";:PRINT USING "\ \";V8$
370 PRINT "q"
380 PRINT #1,"*FREQ:: ";F;" MHz; STP:: ";ST;" MHz; VCO :: YIG"
390 'PRINT #1,"":3"
400 PRINT #1,USING "\ \";V1$;:PRINT #1," ";
410 PRINT #1, USING "\ \";V2$;
420 PRINT #1," ";:PRINT #1,USING "\ \";V3$;
430 PRINT #1," ";:PRINT #1,USING "\ \";V4$
440 PRINT #1,USING "\ \";V5$;:PRINT #1," ";
450 PRINT #1,USING "\ \";V6$;
460 PRINT #1," ";:PRINT #1,USING "\ \";V7$;
470 PRINT #1," ";:PRINT #1,USING "\ \";V8$
480 PRINT #1,"q"
490 CLOSE #1
500 SYSTEM
```